Some Complex Arithmetic and Plotting

Note: Select from the menus "Edit->Execute->Worksheet" (or Ctrl+Shift+Enter) to display results.

```
> # Clear memory, include the "plots" package, and make Maple assume
  that x and y are real variables.
  restart;
  with(plots):

  assume(x,'real');
  assume(y,'real');
```

Here are some basic operations for manipulating complex numbers in Maple.

Keep in mind that Maple uses "I" for the imaginary unit. Also, "abs(z)" computes the modulus of z: "|z|". Most commands are fairly self-explanatory.

```
> conjugate(2+3*I);
  Re(2+3*I);
  Im(2+3*I);
```

$$2 - 3\,I$$
$$2$$
$$3 \qquad \textbf{(1)}$$

```
> abs(1+sqrt(3)*I);
  argument(1+sqrt(3)*I);
```

$$2$$
$$\frac{\pi}{3} \qquad \textbf{(2)}$$

```
> exp(Pi+3*Pi/2*I);
  sin(3*I);
```

$$-e^{\pi + \frac{I\pi}{2}}$$
$$I\sinh(3) \qquad \textbf{(3)}$$

This code generates a plot of a disk of radius R centered at the origin in the complex plane. It also generates a grid of
horizontal and vertical lines (i.e. lines with constant imaginary or constant real parts). N determines the number of gridlines
to generate.

Horizontal lines are paramterized by: x = t and y = some fixed number (= -R+inc*k). In other words, z = t + (-R+inc*k)I.
The bounds for t depend on how high up or down we are in the disk: -sqrt(R^2-(-R+inc*k)^2) <= t <= sqrt(R^2-(-R+inc*k)^2).

Similarly for veritcal lines.

The plot stored in "c" plots the circular edge of the disk. We use polar coordinates to get this parameterization: x = R cos(t) and y = R sin(t).

```
> R := Pi/2;
```

```
N := 10;

inc := R/N:
p := [seq(0,k=0..2*N)]:
q := [seq(0,k=0..2*N)]:
for k from 0 to 2*N do:
   p[k+1] := plot([-R+inc*k,t,t=-sqrt(R^2-(-R+inc*k)^2)..sqrt(R^2-(-R+
inc*k)^2)],color=red);
   q[k+1] := plot([t,-R+inc*k,t=-sqrt(R^2-(-R+inc*k)^2)..sqrt(R^2-(-R+
inc*k)^2)],color=blue);
end do:

c := plot([R*cos(t),R*sin(t),t=0..2*Pi],color=green):

display(p,q,c,scaling=constrained,labels=["Real","Imaginary"]);
```

$$R := \frac{\pi}{2}$$

$$N := 10$$




```
> # Define a complex function.
f := z -> exp(z):
'f(z)' = f(z);

# Get formulas for the real and imaginary parts.
'Re(f(x+y*i))' = simplify(Re(f(x+y*I)));
'Im(f(x+y*i))' = simplify(Im(f(x+y*I)));
```

```
# The radius of the disk we will map using f(z).
R := Pi/2;

# Use a grid of (2N+1)x(2N+1) lines.
N := 10;

inc := R/N:
p := [seq(0,k=0..2*N)]:
q := [seq(0,k=0..2*N)]:
for k from 0 to 2*N do:
   p[k+1] := plot([Re(f(-R+inc*k+t*I)), Im(f(-R+inc*k+t*I)),
                   t=-sqrt(R^2-(-R+inc*k)^2)..sqrt(R^2-(-R+inc*k)^2)],
color=red);
   q[k+1] := plot([Re(f(t+(-R+inc*k)*I)), Im(f(t+(-R+inc*k)*I)),
                   t=-sqrt(R^2-(-R+inc*k)^2)..sqrt(R^2-(-R+inc*k)^2)],
color=blue);
end do:

c := plot([Re(f(R*cos(t)+R*sin(t)*I)),Im(f(R*cos(t)+R*sin(t)*I)),t=0.
.2*Pi],color=green):

display(p,q,c,scaling=constrained,title="Image of Disk",labels=
["Real","Imaginary"]);

# Plot the real part, imaginary part, modulus, and principal argument
of f(z) over the disk |z| <= R.
# plot3d is using a parametric plot:
#       x = r cos(theta), y = r sin(theta), z = Re(f(r cos(theta) + r
sin(theta)I)).
# where 0 <= r <= R and 0 <= theta <= 2Pi.
# We are more or less using polar/cylindrical coordinates to get a
cleaner looking plot (over a disk).

plot3d([r*cos(theta),r*sin(theta),Re(f(r*cos(theta)+r*sin(theta)*I))],
r=0..R,theta=0..2*Pi,scaling=constrained,title="Real Part",labels=
["Real","Imaginary","Re(f)"]);

plot3d([r*cos(theta),r*sin(theta),Im(f(r*cos(theta)+r*sin(theta)*I))],
r=0..R,theta=0..2*Pi,scaling=constrained,title="Imaginary Part",
labels=["Real","Imaginary","Im(f)"]);

plot3d([r*cos(theta),r*sin(theta),abs(f(r*cos(theta)+r*sin(theta)*I))
],r=0..R,theta=0..2*Pi,scaling=constrained,title="Modulus Map",labels=
["Real","Imaginary","|f|"]);

plot3d([r*cos(theta),r*sin(theta),argument(f(r*cos(theta)+r*sin(theta)
*I))],r=0..R,theta=0..2*Pi,scaling=constrained,title="Argument Map",
labels=["Real","Imaginary","Arg(f)"]);
```

$$f(z) = e^{z}$$

$$\Re(f(y\,i + x)) = e^{x\sim} \cos(y\sim)$$

$$\Im(f(i\,y + x)) = e^{x\sim} \sin(y\sim)$$

$$R := \frac{\pi}{2}$$

$$N := 10$$


Image of Disk
Imaginary
2
1
0
−1
−2
1
2
3
4
Real

Real Part



Imaginary Part



Modulus Map



Argument Map




```
> # Same as above but with a different f(z) (this time f(z)=z^2).

  f := z -> z^2:
  'f(z)' = f(z);

  'Re(f(x+y*i))' = simplify(Re(f(x+y*I)));
  'Im(f(x+y*i))' = simplify(Im(f(x+y*I)));

  R := Pi/2;

  N := 10;

  inc := R/N:
  p := [seq(0,k=0..2*N)]:
  q := [seq(0,k=0..2*N)]:
  for k from 0 to 2*N do:
     p[k+1] := plot([Re(f(-R+inc*k+t*I)), Im(f(-R+inc*k+t*I)),
                    t=-sqrt(R^2-(-R+inc*k)^2)..sqrt(R^2-(-R+inc*k)^2)],
  color=red);
     q[k+1] := plot([Re(f(t+(-R+inc*k)*I)), Im(f(t+(-R+inc*k)*I)),
                    t=-sqrt(R^2-(-R+inc*k)^2)..sqrt(R^2-(-R+inc*k)^2)],
  color=blue);
  end do:

  c := plot([Re(f(R*cos(t)+R*sin(t)*I)),Im(f(R*cos(t)+R*sin(t)*I)),t=0.
  .2*Pi],color=green):
```

```
display(p,q,c,scaling=constrained,title="Image of Disk",labels=
["Real","Imaginary"]);

plot3d([r*cos(theta),r*sin(theta),Re(f(r*cos(theta)+r*sin(theta)*I))],
r=0..R,theta=0..2*Pi,scaling=constrained,title="Real Part",labels=
["Real","Imaginary","Re(f)"]);

plot3d([r*cos(theta),r*sin(theta),Im(f(r*cos(theta)+r*sin(theta)*I))],
r=0..R,theta=0..2*Pi,scaling=constrained,title="Imaginary Part",
labels=["Real","Imaginary","Im(f)"]);

plot3d([r*cos(theta),r*sin(theta),abs(f(r*cos(theta)+r*sin(theta)*I))
],r=0..R,theta=0..2*Pi,scaling=constrained,title="Modulus Map",labels=
["Real","Imaginary","|f|"]);

plot3d([r*cos(theta),r*sin(theta),argument(f(r*cos(theta)+r*sin(theta)
*I))],r=0..R,theta=0..2*Pi,scaling=constrained,title="Argument Map",
labels=["Real","Imaginary","Arg(f)"]);
```

$$f(z) = z^2$$

$$\Re(f(iy + x)) = x\sim^2 - y\sim^2$$

$$\Im(f(iy + x)) = 2\,x\sim y\sim$$

$$R := \frac{\pi}{2}$$

$$N := 10$$



Real Part



Imaginary Part



Modulus Map



Argument Map



We redo the above graphs but now instead of using a disk as our domain we use a rectangle.

Our rectangle is ReA <= x <= ReB and ImA <= y <= ImB.

M and N determine the number of horizontal and vertical gridlines drawn.

```
> ReA := -1;
  ReB := 1;
  ImA := 0;
  ImB := Pi;

  M := 10;
  N := 10;

  ReDelta := (ReB-ReA)/M:
  ImDelta := (ImB-ImA)/M:

  p := [seq(0,k=0..M)]:
  q := [seq(0,k=0..N)]:
  for k from 0 to M do:
     p[k+1] := plot([ReA+ReDelta*k,t,t=ImA..ImB],color=red);
  end do:
  for k from 0 to N do:
     q[k+1] := plot([t,ImA+ImDelta*k,t=ReA..ReB],color=blue);
```

```
end do:

display(p,q,scaling=constrained,labels=["Real","Imaginary"]);
```

$$ReA := -1$$
$$ReB := 1$$
$$ImA := 0$$
$$ImB := \pi$$
$$M := 10$$
$$N := 10$$




```
> f := z -> exp(z):
  'f(z)' = f(z);

  'Re(f(x+y*i))' = simplify(Re(f(x+y*I)));
  'Im(f(x+y*i))' = simplify(Im(f(x+y*I)));

  ReA := -1;
  ReB := 1;
  ImA := 0;
  ImB := Pi;

  M := 10;
  N := 10;

  ReDelta := (ReB-ReA)/M:
```

```
ImDelta := (ImB-ImA)/M:

p := [seq(0,k=0..M)]:
q := [seq(0,k=0..N)]:
for k from 0 to M do:
   p[k+1] := plot([Re(f(ReA+ReDelta*k+t*I)),Im(f(ReA+ReDelta*k+t*I)),
t=ImA..ImB],color=red);
end do:
for k from 0 to N do:
   q[k+1] := plot([Re(f(t+(ImA+ImDelta*k)*I)),Im(f(t+(ImA+ImDelta*k)*
I)),t=ReA..ReB],color=blue);
end do:

display(p,q,title="Image of Rectangle",scaling=constrained,labels=
["Real","Imaginary"]);


# Since we're plotting over a rectangular region, we forego
polar/cylindrical coordinates and just use
# a regular (non-parametric) call of plot3d.

plot3d(Re(f(x+y*I)),x=ReA..ReB,y=ImA..ImB,scaling=constrained,title=
"Real Part",labels=["Real","Imaginary","Re(f)"]);

plot3d(Im(f(x+y*I)),x=ReA..ReB,y=ImA..ImB,scaling=constrained,title=
"Imaginary Part",labels=["Real","Imaginary","Im(f)"]);

plot3d(abs(f(x+y*I)),x=ReA..ReB,y=ImA..ImB,scaling=constrained,title=
"Modulus Map",labels=["Real","Imaginary","|f|"]);

plot3d(argument(f(x+y*I)),x=ReA..ReB,y=ImA..ImB,scaling=constrained,
title="Argument Map",labels=["Real","Imaginary","Arg(f)"]);
```

$$f(z) = e^{z}$$

$$\Re(f(y\,i + x)) = e^{x\sim}\cos(y\sim)$$

$$\Im(f(y\,i + x)) = e^{x\sim}\sin(y\sim)$$

$$ReA := -1$$

$$ReB := 1$$

$$ImA := 0$$

$$ImB := \pi$$

$$M := 10$$

$$N := 10$$


Image of Rectangle
Imaginary
Real
2.5
2
1.5
1
0.5
−2
−1
0
1
2

Real Part



Imaginary Part



Modulus Map



Argument Map




```
> # Same as immediately above but with f(z)=z^2 instead of f(z)=exp(z).

  f := z -> z^2:
  'f(z)' = f(z);

  'Re(f(x+y*i))' = simplify(Re(f(x+y*I)));
  'Im(f(x+y*i))' = simplify(Im(f(x+y*I)));

  ReA := -1;
  ReB := 1;
  ImA := 0;
  ImB := Pi;

  M := 10;
  N := 10;

  ReDelta := (ReB-ReA)/M:
  ImDelta := (ImB-ImA)/M:

  p := [seq(0,k=0..M)]:
  q := [seq(0,k=0..N)]:
  for k from 0 to M do:
     p[k+1] := plot([Re(f(ReA+ReDelta*k+t*I)),Im(f(ReA+ReDelta*k+t*I)),
  t=ImA..ImB],color=red);
  end do:
  for k from 0 to N do:
```

```
   q[k+1] := plot([Re(f(t+(ImA+ImDelta*k)*I)),Im(f(t+(ImA+ImDelta*k)*
I)),t=ReA..ReB],color=blue);
end do:

display(p,q,title="Image of Rectangle",scaling=constrained,labels=
["Real","Imaginary"]);



plot3d(Re(f(x+y*I)),x=ReA..ReB,y=ImA..ImB,title="Real Part",labels=
["Real","Imaginary","Re(f)"]);

plot3d(Im(f(x+y*I)),x=ReA..ReB,y=ImA..ImB,title="Imaginary Part",
labels=["Real","Imaginary","Im(f)"]);

plot3d(abs(f(x+y*I)),x=ReA..ReB,y=ImA..ImB,title="Modulus Map",labels=
["Real","Imaginary","|f|"]);

plot3d(argument(f(x+y*I)),x=ReA..ReB,y=ImA..ImB,title="Argument Map",
labels=["Real","Imaginary","Arg(f)"]);
```

$$f(z) = z^2$$

$$\Re(f(y\,i + x)) = x\sim^2 - y\sim^2$$

$$\Im(f(y\,i + x)) = 2\,x\sim y\sim$$

$$ReA := -1$$

$$ReB := 1$$

$$ImA := 0$$

$$ImB := \pi$$

$$M := 10$$

$$N := 10$$


Image of Rectangle
Imaginary
Real
6
4
2
−2
−4
−6
−8
−6
−4
−2
0

Real Part



Imaginary Part



Modulus Map



Argument Map



These examples show plots of modulus maps of several simple polynomials: z, z^2, z^3, z(z-1), and z^2(z-1).

Notice that close to a simple root (i.e. multiplicity one) the graph looks like a cone. Close to a doubly repeated root the graph looks like a paraboloid. Close to a more highly repeated root we get a kind of flattened out paraboloid
like graph.

```
> f := z -> z:
  'f(z)' = f(z);

  'Re(f(x+y*i))' = simplify(Re(f(x+y*I)));
  'Im(f(x+y*i))' = simplify(Im(f(x+y*I)));

  plot3d([r*cos(theta),r*sin(theta),abs(f(r*cos(theta)+r*sin(theta)*I))
  ],r=0..1,theta=0..2*Pi,title="Modulus Map",labels=["Real","Imaginary",
  "|f|"]);
```

$$f(z) = z$$

$$\Re(f(y\,i + x)) = x\sim$$

$$\Im(f(y\,i + x)) = y\sim$$




```
> f := z -> z^2:
  'f(z)' = f(z);

  'Re(f(x+y*i))' = simplify(Re(f(x+y*I)));
  'Im(f(x+y*i))' = simplify(Im(f(x+y*I)));

  plot3d([r*cos(theta),r*sin(theta),abs(f(r*cos(theta)+r*sin(theta)*I))
  ],r=0..1,theta=0..2*Pi,title="Modulus Map",labels=["Real","Imaginary",
  "|f|"]);
```

$$f(z) = z^2$$

$$\Re(f(y\,i + x)) = x\sim^2 - y\sim^2$$

$$\Im(f(y\,i + x)) = 2\,x\sim y\sim$$




```
> f := z -> z^3:
  'f(z)' = f(z);

  'Re(f(x+y*i))' = simplify(Re(f(x+y*I)));
  'Im(f(x+y*i))' = simplify(Im(f(x+y*I)));

  plot3d([r*cos(theta),r*sin(theta),abs(f(r*cos(theta)+r*sin(theta)*I))
  ],r=0..1,theta=0..2*Pi,title="Modulus Map",labels=["Real","Imaginary",
  "|f|"]);
```

$$f(z) = z^3$$

$$\Re(f(y\,i + x)) = x\sim^3 - 3\,x\sim\,y\sim^2$$

$$\Im(f(y\,i + x)) = 3\,x\sim^2\,y\sim - y\sim^3$$




```
> f := z -> z*(z-1):
  'f(z)' = f(z);

  'Re(f(x+y*i))' = simplify(Re(f(x+y*I)));
  'Im(f(x+y*i))' = simplify(Im(f(x+y*I)));

  plot3d([r*cos(theta),r*sin(theta),abs(f(r*cos(theta)+1/2+r*sin(theta)*
  I))],r=0..1,theta=0..2*Pi,title="Modulus Map",labels=["Real",
  "Imaginary","|f|"]);
```

$$f(z) = z\,(z-1)$$

$$\Re(f(y\,i+x)) = x{\sim}^2 - y{\sim}^2 - x{\sim}$$

$$\Im(f(y\,i+x)) = y{\sim}\,(2\,x{\sim} - 1)$$

Modulus Map




```
> f := z -> z^2*(z-1):
  'f(z)' = f(z);

  'Re(f(x+y*i))' = simplify(Re(f(x+y*I)));
  'Im(f(x+y*i))' = simplify(Im(f(x+y*I)));

  plot3d([r*cos(theta),r*sin(theta),abs(f(r*cos(theta)+1/2+r*sin(theta)*
  I))],r=0..1,theta=0..2*Pi,title="Modulus Map",labels=["Real",
  "Imaginary","|f|"]);
```

$$f(z) = z^2 (z-1)$$

$$\Re(f(y\,i + x)) = x\sim^3 - 3\,x\sim y\sim^2 - x\sim^2 + y\sim^2$$

$$\Im(f(y\,i + x)) = y\sim (3\,x\sim^2 - y\sim^2 - 2\,x\sim)$$



Here we graph the modulus map of f(z) = sin(z) and draw in curves to highlight the image of the real and imaginary axes.

Notice that along the real axis (i.e. blue) we get the graph of "|sin(x)|" and along the imaginary axis (i.e. red) we get the
graph of "|sinh(x)|".

All of the roots of sin(z) lie on the real axis. Also, they are simple roots (i.e. not repeated), so our graph looks like a cone
when we stick close to these roots.

```
> f := z -> sin(z):
  'f(z)' = f(z);

  'Re(f(x+y*i))' = simplify(Re(f(x+y*I)));
  'Im(f(x+y*i))' = simplify(Im(f(x+y*I)));

  p := plot3d(abs(f(x+y*I)),x=-2*Pi..2*Pi,y=-2..2,title="Modulus Map",
  labels=["Real","Imaginary","|f|"]):
  q1 := spacecurve(<x,0,abs(f(x+0*I))>,x=-2*Pi..2*Pi,thickness=3,color=
  blue):
  q2 := spacecurve(<0,y,abs(f(0+y*I))>,y=-2..2,thickness=3,color=red):
  display({p,q1,q2});
```

$$f(z) = \sin(z)$$

$\Re(f(y\,i + x)) = \sin(x\sim)\,\cosh(y\sim)$
$\Im(f(y\,i + x)) = \cos(x\sim)\,\sinh(y\sim)$

Modulus Map